


A Tiny BASIC Star Trek program that will have you zapping Klingons till Stardate 29.35



The main object of this Startrek game is to destroy all the Klingon battle cruisers in the Galaxy. The Klingons, Stars and Starbases are randomly positioned within a simulated Galaxy which is divided into a matrix of 8 by 8 sectors, making 64 in all. The Federation starship Enterprise, of which you are in command, has been assigned the task of seeking out and destroying all the Klingon invaders. You have at your disposal an impressive arsenal of weapons, including phasers and photon torpedos which are under your direct control. The Enterprise may be replenished with energy and weapons by docking at a Starbase.

Commands

1) The HELM — This allows you to move the Enterprise through space to any point in your present sector, or to any other sector in the Galaxy. To do this you have to give a warp factor and a course which can be in any one of eight directions. At the end of each manoeuvre you are automatically given a short range scan.


- (a) COURSE (0-7), see Figure. 1
- (b) WARP (1-63), gives you control over the number of units travelled, each unit being 1/8 of a sector.

NOTES (i) To move the Enterprise into another sector you merely give the Helm sufficient warp factor to get there, (each sector is 8 units wide).

(ii) You may not travel through Stars, Klingons or Starbases.

(iii) To dock at a Starbase you manoeuvre into an adjacent position to the left or right of the Starbase.

Fig.1 The eight ~~warp~~ directions.



2) LONG RANGE SCAN — This gives you a summary of the objects in your present sector and adjacent ones. The information is displayed as a three figured number with the hundreds indicating the number of Klingons, the tens the number of Starbases and the units the number of Stars, as shown in Fig. 2.



3) PHASERS — This is an energy weapon, the affect of which diminishes with distance. You are informed of the amount of energy left in the main banks and asked to enter the amount of energy to be diverted to the phaser weapon. To destroy a Klingon you have to deplete his energy to zero. On entering a sector, each Klingon has a full energy bank of 200 units. To calculate the effect of the phasers, the energy is divided equally between the Klingons in the current sector and is further reduced by dividing by the distance between the Klingon and the Enterprise. The result of this calculation is subtracted from the Klingon's energy bank.

Any Klingons left in the sector, after the operation of the weapon, shoot back. The hits on the Enterprise are calculated from the amount left in each Klingon's energy bank, divided by the distance between the Enterprise and the Klingon. However, if you are docked at a Starbase you are protected by its shields.

4) PHOTON TORPEDOS — These are a torpedo like weapon where a single bolt of intense energy is propagated in a single direction. Due to the high energy content of the bolt a Klingon or Starbase is destroyed by a single shot. Stars, however, will absorb the energy. The bolts have an unlimited range within the current sector and are controlled only by direction. The direction command is the same format as that used in the HELM control: COURSE (0-7).

NOTES (i) You cannot shoot through Stars.

(ii) If you destroy a Starbase you are relieved of duty.

(iii) Any Klingons left after your attack will shoot back, using the same procedure as with the phaser weapon.

5) SHIELDS — The shields protect the Enterprise from enemy weaponry. Every hit on the Enterprise depletes the shields by the amount equal to the hit. To protect the Enterprise there must be enough energy in the shields to neutralise any attack or it will be destroyed.

STAR TREK

After requesting the command you are given the total energy available, (the sum of the main banks plus that in the shields). You are then asked to enter the amount of energy to be diverted to the shields.

- NOTES (i) If the command reduces the energy in the shields, any surplus is diverted to the main banks.
 (ii) When docked at a Starbase you are protected by the Starbase shields.
 (iii) As well as being attacked when you shoot at the Klingons, you may also be attacked on a random basis, if the condition is red and you are executing the HELM, SHORT RANGE SCAN or LONG RANGE SCAN commands.

6) SHORT RANGE SCAN – This gives you the positions of the Stars, Klingons, Starbases and the Enterprise in your present sector. You are also given the stardate, the status of the main energy banks, your condition, (GREEN, RED, DOCKED), the number of photon torpedos left, the shield energy and the total number of Klingons left in the Galaxy.

7) RESIGNATION . . .

Video Display Characters

The objects in the Galaxy are displayed as follows:–

Object	Displayed Character
Enterprise	-0-
Klingon	> <
Star	*
Starbase	<O>

Comments on Program Conversion for other Computers

- 1) The IF statement operates on the rest of the line and can be used to control multiple statements.
- 2) The N in the PRINT statements control the number of digits printed.
- 3) The RND(N) function generates a random number between 1 and N.
- 4) The TRITON variables are two byte integers which is the minimum length for the program's operation.

Program Notes

Line Nos.	Description
20 – 90	Initialisation of variables.
100 – 150	Initialisation of sectors. Each sector is represented as a single number – the hundreds representing the number of Klingons, the tens the number of Starbases and the units the number of Stars.
160	Choose the initial current sector.
170	Choose the initial position of the Enterprise in the current sector.
180 – 190	Joins up the ends of the Galaxy.
200 – 330	Sets up the positions of the Stars, Starbases and Klingons in the current sector.
200 – 220	Decodes the sector number.
230	Zeros sector and loads Enterprise position.
245 – 260	Sets up the Stars.
270 – 290	Sets up the Starbases.

300 – 320	Sets up the Klingons.
360 – 370	Tests for Enterprise condition – GREEN, RED, DOCKED.
380	Reloads weapons and energy banks when condition DOCKED.
400 – 540	Short Range scan outputs the position of the Stars, Starbases, Klingons and the Enterprise; also gives information on the stardate, energy banks, shields, torpedos, total number of Klingons in the Galaxy, etc.
550	Random test for Klingon to fire on Enterprise after a command.
600	Test for all Klingons destroyed.
605 – 620	Select command (1–7)
1000 – 1140	HELM control (1)
1000 – 1035	Read warp factor.
1045	Blank Enterprise position in the current sector matrix: decrement energy and increment date.
1090 – 1130	Move Enterprise W units.
1135	Calculate new sector if Enterprise leaves the current sector.
1140	Put new Enterprise position into the current sector matrix.
2000 – 2040	LONG RANGE SCANE (2) Prints out the sector numbers of the current sector and the surrounding ones.
3000 – 3120	PHASER control (3)
3000 – 3010	Prints energy reserve and reads phaser energy.
3020	Decrement energy banks, and test for no Klingons in current sector.
3030	Divide phaser energy by the number of Klingons in the current sector and calculate the Enterprise position vector.
3040 – 3050	Search current sector matrix for Klingons.
3060 – 3070	Calculate distance between Enterprise and Klingons: Decrement Klingon energy bank.
3080	Test for Klingon destroyed and take appropriate action.
3090	Klingon not destroyed so store new Klingon energy level: Calculate hit on Enterprise and subtract from shields.
3110	Test for destruction of the Enterprise.
3120	Updates the number of Klingons in the current sector.
4000 – 4080	PHOTON TORPEDOS control (4)
4000	Test for no torpedos
4010	Read course: decrement number of torpedos.
4010 – 4050	Step along course until object encountered.
4060	Test for Starbase hit.
4065	Test for object not a Klingon.
4070	Klingon destroyed.
4080	Set phaser energy to zero and jump to phaser routine to calculate the hits on the Enterprise shields from the remaining Klingons.
5000 – 5030	SHIELD control (5)
5000	Add shield power to that in the main energy banks: Prints total energy.
5010	Read shield energy and subtract it from the main energy banks.
5020	Jump for next command.
6000	SHORT RANGE SCAN (6) Jump to short range scan routine.
7000	RESIGNATION (7)
8000 – 8030	Subroutine to calculate distance between the Enterprise and Klingons.
8200	Klingon destroyed subroutine.

8300 Print energy subroutine.
8500 - 8570 Subroutine to read course and calculate the direction increment vector.
8600 Game lost: Stop.

NOTE
The subscripted characters in the PRINT statements used to display the Enterprise, etc., (line nos. 440 - 460) indicate the character is to be typed in conjunction with the control or shift key.

Fig.3 The program listing.

```

10 REM STARTREK SIMULATION PROGRAM
20 REM A TINY BASIC VERSION
30 REM I.L.POWELL 10/3/1979
40 K=0
50 T=RND(200)+200
60 E=3000
70 O=15
80 S=0
90 L=T
95 REM SET UP SECTORS
100 FOR I=0 TO 63
103 X=0
105 Y=0
110 IF RND(10)<8 X=RND(3)
120 IF RND(100)>88 Y=1
130 Z=RND(5)
140 @(I)=X*100+Y*10+Z
145 K=K+X
150 NEXT I
160 Q=RND(64)-1
170 A=RND(64)-1
180 IF Q<0 Q=Q+64
190 IF Q>63 Q=Q-64
199 REM SET UP CURRENT SECTOR
200 Z=@(Q)
210 X=Z/100
215 Z=Z-X*100
220 Y=Z/10
225 Z=Z-Y*10
230 FOR I=64 TO 127
233 @(I)=0
235 NEXT I
237 @(A+64)=4
240 FOR I=1 TO 5
245 IF I>Z GOTO 270
250 J=RND(64)+63
255 IF @(J)#0 GOTO 250
260 @(J)=3
270 IF I>Y GOTO 300
280 J=RND(64)+63
285 IF @(J)#0 GOTO 280
290 @(J)=2
300 IF I>X GOTO 330
310 J=RND(64)+63
315 IF @(J)#0 GOTO 310
320 @(J)=-200

```

```

330 NEXT I
340 REM SHORT RANGE SCAN
350 C=1
355 IF X#0 C=2
370 IF (@(A+63)=2)+(@(A+65)=2) C=3
380 IF C=3 E=3000; S=0; O=15
400 FOR I=0 TO 56 STEP 8
410 FOR J=64 TO 71
420 U=@(I+J)
430 IF U=0 PRINT ' ',
440 IF U<0 PRINT '>:<',
450 IF U=2 PRINT '<0>',
455 IF U=3 PRINT '* ',
460 IF U=4 PRINT '-0-',
470 NEXT J
480 IF I=8 PRINT 'STARDATE',T
485 IF (I=0)+(I=56) PRINT
490 IF I#16 GOTO 500
492 PRINT 'CONDITION',
493 IF C=1 PRINT 'GREEN'
494 IF C=2 PRINT 'RED'
496 IF C=3 PRINT 'DOCKED'
500 IF I=24 PRINT 'ENERGY',E
510 IF I=32 PRINT 'TORPEDOS',O
520 IF I=40 PRINT 'SHIELDS',S
530 IF I=48 PRINT 'KLINGONS',K
540 NEXT I
550 IF L#T IF RND(9)<6 GOTO 4080
590 REM COMMAND CONTROL
600 IF K=0 PRINT 'THE FEDERATION HAS BEEN SAVED'; STOP
605 INPUT 'COMMAND' B
610 IF (B>7)+(B<1) GOTO 600
620 GOTO B*1000
999 REM HELM CONTROL
1000 GOSUB 8500
1020 INPUT 'WARP(1-63)'W
1030 IF (W<1)+(W>69) GOTO 1020
1035 IF E W GOSUB 8300; GOTO 600
1040 @(A+64)=0
1045 E=E-W
1050 T=T+W
1090 FOR I=1 TO W
1110 IF (V<0)+(V>7)+(H<0)+(H>7) GOTO 1120
1113 U=0
1115 IF @(V+8*H+64)#0 V=V-N; H=H-M; GOTO 1140
1120 NEXT I
1125 IF V<0 V=V-8
1130 IF H<0 H=H-8
1135 IF U Q=Q+8*(H/8)+V/8; GOTO 170
1140 A=V+8*H
1150 @(A+64)=4

```

RELOADS WHEN DOWNED (C=3)

Multiple hit

STAR TREK

```

1160 GOTO 350
1999 REM LONG RANGE SCAN
2000 FOR I=-8 TO 8 STEP 8
2010 FOR J=-1 TO 1
2015 U=Q+J+I
2020 IF U<0 U=U+64
2025 IF U>63 U=U-64
2030 PRINT#4,@(U),
2033 NEXT J
2035 PRINT
2037 NEXT I
2040 GOTO 550
2999 REM PHASER CONTROL
3000 GOSUB 8300
3005 INPUT 'PHASERS' P
3010 IF E<P GOTO 3000
3020 E=E-P
3025 IF X=0 GOTO 3120
3030 P=P/X
3033 H=A/8
3035 V=A-8*H
3040 FOR I=0 TO 63
3050 J=I+64
3055 IF @(J)>=0 GOTO 3110
3060 F=I/8
3065 U=I-F*8
3070 GOSUB 8000
3073 G=@(J)
3075 G=G+P/D
3080 IF G>=0 GOSUB 8200
3085 GOTO 3110
3090 @(J)=G
3100 IF C#3 G=-G/D ; S=S-G ;
PRINT#3,G,'HIT ON SHIELDS'
3110 NEXT I
3115 IF S<0 GOTO 8100
3120 @(Q)=(Q)-(@(Q)/100-X)*100
3130 GOTO 600
3999 REM PHOTON TORPEDO CONTROL
4000 IF O<1 PRINT 'NO TORPEDOS' ;
GOTO 600
4010 GOSUB 8500
4013 O=O-1
4015 FOR I=1 TO 16
4020 V=V+N
4025 H=H+M
4030 IF (V<0)+(V>7)+(H<0)+(H>7)
GOTO 4080
4040 J=V+8*H+64
4045 IF @(J)#0 GOTO 4060
4050 NEXT I
4060 IF @(J)=2 PRINT 'STAR BASE
DESTROYED' ; GOTO 8600
4065 IF @(J)>0 GOTO 4080
4070 GOSUB 8200

```

```

4080 P=0
4090 GOTO 3020
5000 E=E+S
5005 GOSUB 8300
5010 INPUT 'SHIELDS' S
5015 E=E-S
5020 IF E<1 PRINT 'S>E' ;
GOTO 5000
5030 GOTO 600
5999 REM JUMP TO SHORT
RANGE SCAN
6000 GOTO 350
6999 REM RESIGNATION
7000 PRINT 'ACCEPTED'
7010 GOTO 8600
7900 REM SUBROUTINES
7910 REM CALCULATE KLINGON
ENTERPRISE DISTANCE
8000 Z=H-F
8005 Y=V-U
8010 FOR D=1 TO 8
8015 G=D+1
8020 IF G*G (Z*Z+Y*Y) RETURN
8030 NEXT D
8040 RETURN
8090 REM
8100 PRINT 'ENTERPRISE DEAD
IN SPACE'
8110 GOTO 8600
8190 REM
8200 PRINT 'KLINGON DESTROYED'
8210 X=X-1
8220 K=K-1
8230 @(J)=0
8240 RETURN
8290 REM
8300 PRINT 'ENERGY=',#5,E
8310 RETURN to 1000
8390 REM READ COURSE AND
CALCULATE MOVEMENT VECTOR
8500 INPUT (COURSE(0-7)) B
8510 IF (B<0)+(B>7) GOTO 8500
8520 M=0
8523 N=0
8525 H=A/8
8527 V=A-H*8
8530 IF (B<2)+(B>6) M=-1
8540 IF B>2 IF B<6 M=1
8550 IF B<4 IF B>0 N=1
8560 IF B>4 N=-1
8570 RETURN to 1000
8590 REM
8600 PRINT 'YOU ARE RELIEVED
OF DUTY'
8610 STOP

```